Mirosława Kopertowska
Witold Sikorski

Funkcje w Excelu w praktyce

Wydanie II

PWN

Redakcja **Matylda Pawłowska**

Skład komputerowy **Witold Świstak**

Zastrzeżonych nazw firm i produktów użyto w książce wyłącznie w celu identyfikacji

ISBN 978-83-01-15831-6

Wydanie II – 4 dodruk

Wydawnictwo Naukowe PWN SA
tel. 22 69 54 321; faks 22 69 54 288
infolinia 801 33 33 88
e-mail: pwn@pwn.com.pl
www.pwn.pl

Spis treści

Wprowadzenie

Funkcje spełniają w arkuszu kalkulacyjnym Excel coraz większą rolę. Dawno już wyszły poza zestaw podstawowych funkcji matematycznych, obejmując coraz szerszy zakres. Celem ćwiczeń jest pokazanie zastosowania poszczególnych funkcji na przykładach praktycznych. Często opis funkcji w menu i w pomocy jest dość lakoniczny i dopiero praktyczne przykłady mogą uzmysłowić użytkownikom takie ich zastosowania, których często nawet nie podejrzewali.

Korzystanie z funkcji

Funkcje mogą być używane w arkuszu kalkulacyjnym wyłącznie jako elementy formuł (wzorów matematycznych) niezależnie od tego, czy występują w ramach większego wyrażenia, czy też samodzielnie. Tak więc zapis w komórce *SUMA(A1:A4)* nie wywoła żadnego efektu. Zostanie potraktowany jako tekst i w takiej postaci pojawi się na ekranie. Wynik sumowania pojawi się w komórce, gdy wpiszemy do niej wzór *=SUMA(A1:A4)*. Wtedy arkusz potraktuje zapis jako formułę i wykona działanie. Wpisanie błędnej nazwy, np. *=SUMKA(A1:A4)*, spowoduje pojawienie się komunikatu o błędzie. Błędy omówiono oddzielnie w **dodatku C**.

Funkcję zapisujemy, podając jej nazwę i listę argumentów, które są umieszczone w nawiasach okrągłych. Liczba i postać argumentów zależą od typu funkcji. Argumentami mogą być nazwy komórek lub zakresów komórek bądź stałe odpowiedniego typu. Jeśli funkcja nie ma argumentów, i tak musi mieć nawiasy – nie obejmują wówczas żadnej wielkości (np. *TERAZ()*). Jeśli funkcja ma więcej niż jeden argument, są oddzielane średnikami – znak ***;***.

Uwaga!

Można stosować inny znak oddzielający argumenty. Określa się go w **Panelu sterowania** w **Ustawieniach regionalnych**. Na karcie **Liczby** znajduje się opcja **Separator listy**, która określa separatory występujące w programach uruchamianych w systemie Windows. W Polsce standardowym ustawieniem jest średnik. Jednak trzeba pamiętać, że w USA jest to przecinek, a ustawienie może być dowolnie zmienione.

W wyrażeniach zawierających funkcje mogą oczywiście występować działania odpowiedniego typu. W **dodatku B** podano informacje na temat typów działań, stosowanych operatorów oraz ich hierarchii.

Rodzaje funkcji

Funkcje dostępne w Excelu podzielono na kilka grup. Podział ten nie jest sztywny i w kolejnych wersjach programu występują pewne różnice. Oto dziewięć podstawowych grup w kolejności przyjętej przez twórców programu:

- finansowe – funkcje dotyczące kapitału, odsetek, amortyzacji, inwestycji kapitałowych itp. (nazwy funkcji nie zostały spolszczone);
- daty i czasu – funkcje dotyczące daty, godziny, dni tygodnia oraz obliczeń związanych z upływem czasu;
- matematyczne – funkcje trygonometryczne, hiperboliczne, logarytmiczne, sumowania, zaokrąglenia oraz pewne działania macierzowe;
- statystyczne – funkcje związane z obliczeniami z zakresu rachunku prawdopodobieństwa i statystyki, w tym odchylenia, rozkłady statystyczne oraz testy zgodności;
- wyszukiwania i adresu – funkcje związane z położeniem komórek w arkuszu lub odwołaniem do adresów w sieci;
- bazy danych – funkcje związane z obliczeniami na obszarach arkusza zdefiniowanych jako bazy danych;
- tekstowe – funkcje związane z wyszukiwaniem i przekształcaniem tekstów;
- logiczne – sześć podstawowych funkcji logicznych;
- informacyjne – informacje na temat typu zawartości komórek oraz poprawności wykonania formuł.

Kolejne rozdziały obejmują ćwiczenia dotyczące funkcji z poszczególnych grup. Niektóre funkcje omawiane są zbiorczo – na przykład nie ma ćwiczeń dotyczących każdej dostępnej funkcji trygonometrycznej. Autorzy dokonali wyboru funkcji, więcej miejsca poświęcając częściej stosowanym narzędziom. Przyjęta w książce kolejność omawiania grup funkcji związana jest z ich stopniem komplikacji i częstością stosowania w praktyce. Nie będą omawiane wszystkie funkcje w kolejności alfabetycznej, niektóre działy są zaś bardziej rozbudowane od innych. Autorzy starali się odrzucić kryteria alfabetyczne, nie chcąc powtarzać podręczników do Excela oraz informacji dotyczących pomocy.

Książka ma charakter ćwiczeń i nie jest jej celem powtarzanie opisów funkcji znajdujących się w licznych książkach i w pomocy. Dlatego też zestawienie liczby i typów argumentu każdej funkcji umieszczono jedynie na końcu książki w **dodatku A**. Wymieniono tam także funkcje niewystępujące w przykładach.

Uwaga!

W kolejnych polskich wersjach programu występują różnice w przekładzie nazw funkcji. Autorzy opierali się na nazwach stosowanych w wersji 2003. Tam, gdzie istnieją różnice, zostało to zaznaczone.

Na zakończenie tego krótkiego wprowadzenia warto dodać, że w Excelu istnieje możliwość tworzenia własnych funkcji. Funkcje te znane jako funkcje definiowane przez użytkownika są tworzone za pomocą języka Visual Basic dla Aplikacji. Temat ten został dokładnie omówiony w książce Julitty Korol „Visual Basic dla Aplikacji w Excelu 2000” wydanej przez Wydawnictwo MIKOM.

Stosowane konwencje

W książce przyjęto, że nazwy funkcji będą pisane wielkimi literami, choć nie jest to wymagane w Excelu. Argumenty podawane są małymi literami. To wystarczy do wyróżnienia ich w tekstach ćwiczeń. Także oznaczenia kolumn w arkuszu podawane są wielkimi literami.

Wzory, wyrażenia i nazwy zmiennych w tekście ćwiczeń pisane są czcionką `courier`. Ćwiczenia ilustrowane są wybranymi fragmentami arkusza kalkulacyjnego, aby uniknąć zbędnego podawania całych ekranów. Pozwala to także uniezależnić ilustracje od wersji programu. Tam, gdzie w przykładzie występuje niewiele formuł matematycznych, są one opisane w odnośnikach do komórek tak, jak pokazano poniżej:

	A	B	C	D	E	F	G	H	I
1	2,50	-1,00	0,10	**1,60**					
2	0,80	0,00	-1,20	**-0,40**					
3	-21,00	2,55	13,50	**-4,95**					
4	0,05	-12,80	12,00	**-0,75**					
5	-0,03	28,50	-4,56	**23,91**					
6				**1 779,98** ←	=SUMA.KWADRATÓW(A1:C5)				

a w przypadku wielu formuł rysunki składają się z dwóch części: pierwsza pokazuje wyniki, a druga wszystkie formuły:

	A	B	C
1		dane w radianach	wynik w stopniach
2	**π**	3,141592654	180
3	**π/4**	0,785398163	45
4			
5		wynik w stopniach	dane w radianach
6		90	1,570796327
7		180	3,141592654
8		-45	-0,785398163

	A	B	C
1		dane w radianach	wynik w stopniach
2	**π**	=PI()	=STOPNIE(B2)
3	**π/4**	=B2/4	=STOPNIE(B3)
4			
5		wynik w stopniach	dane w radianach
6		90	=RADIANY(B6)
7		180	=RADIANY(B7)
8		-45	=RADIANY(B8)

.

Czytelnik może się przełączać między widokami wyników i formuł za pomocą polecenia **Narzędzia/Opcje** – w zakładce **Widok** należy odpowiednio wyłączyć lub włączyć pole wyboru **Formuły**:



.

Uwagi zawierające informacje, o których warto pamiętać, umieszczono w ramkach.

1. Podstawowe funkcje matematyczne

Stosowanie funkcji w arkuszach kalkulacyjnych zaczęło się od funkcji matematycznych. Wiele tego rodzaju funkcji (pierwiastek, logarytm, funkcje trygonometryczne) jest standardowo dostępnych w większości języków programowania i w naturalny sposób znalazły się w pierwszych arkuszach kalkulacyjnych, a więc i w Excelu. Dlatego też od nich zaczniemy przykłady zastosowania funkcji.

Funkcje matematyczne służą do wyznaczania najczęściej używanych wartości. Mogą być wykorzystywane w dowolnie skomplikowanych wzorach matematycznych. Argumenty tych funkcji wynikają z matematycznych definicji poszczególnych funkcji.

Funkcje trygonometryczne i hiperboliczne

Funkcje trygonometryczne i hiperboliczne obejmują zarówno funkcje podstawowe, jak i odwrotne. Kąty wyrażone są w radianach. Kilka ćwiczeń zaprezentuje poszczególne grupy tych funkcji.

Ćwiczenie 1.1

Oblicz wysokość komina, jeśli wiadomo, że z odległości dl = 60 m widać go pod kątem α = 56 stopni.

Jest to podstawowe zadanie z geometrii, w którym z zależnościami obowiązującymi w trójkącie prostokątnym szukaną wysokość `x` oblicza się według wzoru: `x = dl*tan(α).`

W tym miejscu pojawia się problem wyrażenia argumentu w radianach, czego wymagają funkcje trygonometryczne. Można użyć funkcji RADIANY(k), która przekształca kąt k wyrażony w stopniach na radiany. Tak więc ostateczne wyliczenie szukanej wartości będzie miało postać widoczną na rysunku 1.1.

	A	B	C	D	E	F	G
1	**odległość *dl***	**kąt *α***	**wysokość *x***				
2	60	56	88,95365811 ←	=A2*TAN(RADIANY(B2))			
3							

Rysunek 1.1. Funkcje TAN i RADIANY

Oczywiście we wzorze można użyć zarówno nazw komórek, jak i bezpośrednio podanych wartości. Powyższe rozwiązanie może przyjąć prostszą postać przedstawioną na kolejnym rysunku (1.2).

	A	B	C	D	E	F	G
1	**odległość *dl***	**kąt *α***	**wysokość *x***				
2	60	56	88,95365811				
3							

=60*TAN(RADIANY(56))

Rysunek 1.2. Funkcje TAN i RADIANY z argumentami podanymi w postaci stałych

Wynik jest oczywiście w obu przypadkach taki sam. Wysokość komina wynosi w zaokrągleniu 88,9537 metra. W jednym z kolejnych ćwiczeń pokażemy sposób zaokrąglania wyniku w celu uzyskania oczekiwanej dokładności.

Uwaga!

Argumentem funkcji RADIANY jest wartość wyrażona w stopniach, a wynik wyrażony jest w radianach. Z punktu widzenia obliczeniowego są to w arkuszu „pewne liczby rzeczywiste". Poprawność podanych jednostek zależy wyłącznie od użytkownika i nie jest w żaden sposób sprawdzana przez funkcje. Przeliczenia pokazano na rysunku 1.3.

B2 fx =RADIANY(A2)

	A	B
1	kąt wyrażony w stopniach	kąt wyrażony w radianach
2	56	0,977384381
3	90	1,570796327
4	0	0
5	180	3,141592654

Rysunek 1.3. Przeliczanie stopni na radiany (postać wzoru widać w pasku formuły)

Ćwiczenie 1.2

Porównaj wartość funkcji cosinus hiperboliczny wyznaczoną za pomocą funkcji arkusza COSH oraz na podstawie wzoru.

Podany wzór:

$$\cosh(x) = \frac{e^{x} - e^{-x}}{2}$$

wymaga użycia funkcji EXP, która oblicza wartość liczby Eulera (e) podniesionej do zadanej potęgi *x*.

	A	B	C	D
1		liczba *x*		
2		0	-0,5	1,2
3	wynik funkcji COSH	1	1,127626	1,810656
4	wynik wzoru	1	1,127626	1,810656

	A	B	C	D
1		liczba *x*		
2		0	-0,5	1,2
3	wynik funkcji COSH	=COSH(B2)	=COSH(C2)	=COSH(D2)
4	wynik wzoru	=(EXP(B2)+EXP(-B2))/2	=(EXP(C2)+EXP(-C2))/2	=(EXP(D2)+EXP(-D2))/2

Rysunek 1.4. Wartość cosinusa hiperbolicznego można obliczyć na dwa sposoby

Obie zastosowane w ćwiczeniu funkcje używają jednego argumentu x reprezentującego liczbę całkowitą, której cosinus hiperboliczny obliczamy. Wyniki są identyczne, co świadczy o tym, że obie funkcje wyznaczane są z maksymalną dokładnością dostępną w arkuszu Excel.

Ćwiczenie 1.3

Przekształć liczbę π oraz π/4 na stopnie, a następnie zamień na radiany 90°, 180° oraz –45°. Porównaj otrzymane wyniki.

Liczbę π możemy otrzymać z dostępną dokładnością, korzystając z funkcji PI. Przekształcenie z radianów na stopnie realizuje funkcja STOPNIE, która jest funkcją odwrotną do omówionej wcześniej funkcji RADIANY. Na rysunku pokazano wyniki, a poniżej odpowiadające im formuły.

	A	B	C
1		dane w radianach	wynik w stopniach
2	π	3,141592654	180
3	π/4	0,785398163	45
4			
5		wynik w stopniach	dane w radianach
6		90	1,570796327
7		180	3,141592654
8		-45	-0,785398163

	A	B	C
1		dane w radianach	wynik w stopniach
2	π	=PI()	=STOPNIE(B2)
3	π/4	=B2/4	=STOPNIE(B3)
4			
5		wynik w stopniach	dane w radianach
6		90	=RADIANY(B6)
7		180	=RADIANY(B7)
8		-45	=RADIANY(B8)

Rysunek 1.5. Przekształcenia radiany-stopnie

Jak widać, poprawność przekształcenia w zakresie dokładności obliczeń komputera (11 cyfr znaczących) potwierdza się podczas zamiany z radianów na stopnie i odwrotnie.

Funkcje logarytmiczne, wykładnicze i potęgowe

Podobnie jak w przypadku funkcji trygonometrycznych, w tej grupie funkcji można znaleźć popularne, stosowane na co dzień działania matematyczne. Potęga, pierwiastek, logarytm to podstawowe funkcje stosowane w równaniach algebraicznych.

Ćwiczenie 1.4

Wyznacz logarytm naturalny i dziesiętny podanej liczby oraz jej logarytm przy podstawie 2. Sprawdź działanie za pomocą funkcji potęgowania.

Logarytm naturalny to logarytm obliczany przy podstawie liczby Eulera (*e=2,71828182845904...*), a logarytm dziesiętny ma podstawę dziesięć. Oba logarytmy są często stosowane w praktyce i dlatego odpowiadają im oddzielne funkcje LN i LOG10.

Wyznaczenie logarytmu przy podstawie 2 wymaga użycia funkcji dwuargumentowej LOG, której drugim parametrem jest podstawa logarytmu. Dziedziną funkcji logarytmicznej są liczby rzeczywiste dodatnie, więc argument nie może być liczbą ujemną lub zerem. Rysunek 1.6 pokazuje poprawnie wyznaczane wartości logarytmów oraz odpowiadające im formuły, a rysunek 1.7 – rezultat w przypadku błędnego argumentu.

	A	B	C	D	E
1	liczba x	85	2,718282	10	2
2	ln(x)	4,442651	1	2,302585	0,693147
3	log10(x)	1,929419	0,434294	1	0,301030
4	log(x;2)	6,409391	1,442695	3,321928	1

	A	B	C	D	E
1	liczba x	85	2,71828182845904	10	2
2	ln(x)	=LN(B1)	=LN(C1)	=LN(D1)	=LN(E1)
3	log10(x)	=LOG10(B1)	=LOG10(C1)	=LOG10(D1)	=LOG10(E1)
4	log(x;2)	=LOG(B1;2)	=LOG(C1;2)	=LOG(D1;2)	=LOG(E1;2)

Rysunek 1.6. Przykłady wykorzystania funkcji logarytmicznej

Ponieważ wartość logarytmu z liczby, która jest jego podstawą, wynosi 1, dla sprawdzenia pokazano logarytmy z liczb e, 10 i 2. Jak widać, w zakresie dokładności arkusza wyniki są zgodne z oczekiwaniami. Dla każdego z logarytmów otrzymujemy wartość 1 jako wynik, gdy argumentem jest jego podstawa.

Zgodnie z tym, co wiemy z matematyki, każdy z logarytmów sygnalizuje błąd, gdy jego argument jest mniejszy lub równy zeru. Niewłaściwa wartość argumentu sygnalizowana jest zawsze błędem #LICZBA!, bez określania, w czym tkwi problem.

	A	B	C
1	liczba **x**	0	-5
2	ln(x)	#LICZBA!	#LICZBA!
3	log10(x)	#LICZBA!	#LICZBA!
4	log(x;2)	#LICZBA!	#LICZBA!

Rysunek 1.7. Błędne argumenty dają taki sam wynik dla każdej funkcji

Do sprawdzenia wyników obliczania wartości logarytmów wykorzystamy potęgowanie. W przykładzie obliczono wartości logarytmów z podanej liczby x, a następnie podniesiono podstawę logarytmu do otrzymanej wartości jako potęgi.

	A	B	C	D
1	liczba **x**	85		
2	ln(x)	4,442651256		85
3	log10(x)	1,929418926		85
4	log(x;2)	6,409390936		85

	A	B	C	D
1	liczba **x**	85		
2	ln(x)	=LN(B1)		=EXP(B2)
3	log10(x)	=LOG10(B1)		=POTĘGA(10;B3)
4	log(x;2)	=LOG(B1;2)		=POTĘGA(2;B4)

Rysunek 1.8. Funkcje odwrotne do funkcji logarytmicznych EXP i POTĘGA

Uwaga!
Logarytm naturalny ma własną jednoargumentową funkcję odwrotną EXP. Natomiast funkcja POTĘGA odwrotna dla pozostałych logarytmów ma dwa argumenty – podstawę i wykładnik potęgi.

Ćwiczenie 1.5

Oblicz pierwiastki wielomianu drugiego stopnia $2,4x^2-6,8x+1,2$ *oraz jego wartość* y *w zadanym punkcie* x.

Do obliczenia pierwiastków wielomianu korzystamy ze znanych wzorów, w których musimy użyć pierwiastka kwadratowego. Obliczenie wartości wielomianu wymaga obliczenia x^2, co można zrealizować na dwa sposoby – za pomocą operatora potęgowania ^ lub funkcji POTĘGA. Ta ostatnia pozwala podnieść liczbę dodatnią do dowolnej potęgi rzeczywistej oraz liczbę ujemną do potęgi całkowitej.

	A	B	C
1	**y=2,4*x^2-6,8*x+1,2**		
2	**delta**	34,72	
3		**x1**	**x2**
4	**pierwiastki**	1,101768	0,078788
5	**wartość *y* dla *x=0,5***	-1,60	

	A	B	C
1	**y=2,4*x^2-6,8*x+1,2**		
2	**delta**	=(-6,8)^2-4*2,4*1,2	
3		**x1**	**x2**
4	**pierwiastki**	=((6,8)+PIERWIASTEK(B2))/(4*2,4*1,2)	=-((-6,8)+PIERWIASTEK(B2))/(4*2,4*1,2)
5	**wartość *y* dla *x=0,5***	=2,4*0,5^2-6,8*0,5+1,2	

Rysunek 1.9. Wzór z wykorzystaniem funkcji pierwiastka kwadratowego

Wykorzystanie funkcji we wzorach daje duże możliwości opracowywania różnego rodzaju obliczeń matematycznych. Warto zwrócić uwagę, że pokazane wyżej wyznaczenie wyróżnika równania kwadratowego (delta) obliczanego tu w komórce **B2** na podstawie wzoru pokazanego w komórce **A1** może mieć także inną postać:

POTĘGA(–6,8;2)–ILOCZYN(4;2,4;1,2)

Wykorzystano tu poznaną wcześniej funkcję potęgowania oraz funkcję obliczania iloczynu. Parametry funkcji ILOCZYN omówiono dokładnie przy opisie funkcji SUMA, która działa analogicznie.

Zaokrąglanie i obcinanie wartości liczb

Wśród operacji matematycznych dość istotną rolę pełnią funkcje zaokrąglania wartości. Ma to duże znaczenie w obliczeniach i przedstawianiu wyników działań wykonywanych w arkuszach kalkulacyjnych. Ponieważ istnieje wiele funkcji w różny sposób spełniających tę rolę, na wstępie przedstawione zostaną istniejące możliwości.

Ćwiczenie 1.6

Wynik otrzymany w ćwiczeniu 1.1 należy zaokrąglić, zaokrąglić w dół i zaokrąglić w górę z dokładnością do jednego i dwóch miejsc po przecinku. Wyznacz też najbliższą liczbę całkowitą.

Wszystkie użyte tu funkcje zaokrąglania mają po dwa argumenty. Pierwszym argumentem jest liczba, którą chcemy zaokrąglić, drugą wymagana liczba miejsc po przecinku. Funkcja wyznaczająca najbliższą wartość całkowitą jest jednoargumentowa.

	A	B	C
1	liczba **x**:	88,95365811	
2	zaokrąglenia liczby **x**		
3	do jednego miejsca po przecinku	do dwóch miejsc po przecinku	do wartości całkowitej
4	89,000	88,950	88,000
5	88,900	88,950	
6	89,000	88,960	

	A	B	C
1	liczba **x**:	88,95365811	
2	zaokrąglenia liczby **x**		
3	do jednego miejsca po przecinku	do dwóch miejsc po przecinku	do wartości całkowitej
4	=ZAOKR(B1;1)	=ZAOKR(B1;2)	=LICZBA.CAŁK(B1)
5	=ZAOKR.DÓŁ(B1;1)	=ZAOKR.DÓŁ(B1;2)	
6	=ZAOKR.GÓRA(B1;1)	=ZAOKR.GÓRA(B1;2)	

Rysunek 1.10. Wyniki działania różnych funkcji zaokrąglających

Warto porównać przedstawione na rysunku wyniki. Zależą oczywiście od postaci zaokrąglanej liczby. Gdy usuwana część rozpoczyna się od cyfry mniejszej od 5, to standardowe zaokrąglenie jest zaokrągleniem w dół. W przeciwnym razie funkcja ZAOKR zaokrągla w górę. Natomiast jak wynika z nazwy, ZAOKR.DÓŁ zawsze odrzuca zbędną część liczby. Dokładnie przeciwny efekt osiągniemy przy zaokrąglaniu w górę funkcją ZAOKR.GÓRA.

Przy wyświetlaniu w komórkach wartości liczbowych trzeba pamiętać, że program nie zawsze wyświetla całą informację zawartą w pamięci. Jeśli brak miejsca do wyświetlenia wartości lub zostanie zadany określony format, to na ekranie zobaczymy jej skróconą postać, a efekt wizualny będzie podobny do zaokrąglenia.

Nie wolno mylić tych dwóch sytuacji, gdyż to, co widać na ekranie, nie zawsze jest zgodne z zawartością pamięci. W pamięci wartość nie zmienia się zależnie od sposobu wyświetlania. Zmiana wartości (zaokrąglenie) ma miejsce tylko wtedy, gdy użyjemy jednej z funkcji.

Ćwiczenie 1.7

Porównaj wartość wyświetlaną z dwoma miejscami po przecinku z drugą zaokrągloną w ten sam sposób.

Wynik obliczeń wynoszący dokładnie 88,95365811 wyświetlimy, określając format liczbowy na dwa miejsca po przecinku i zaokrąglimy wartość w ten sam sposób. Następnie zawartość komórek pomnożymy przez dużą liczbę, np. 50 000.

Można zaobserwować, że choć liczby wyświetlone w komórkach **A5** i **B5** wydają się identyczne, wynik mnożenia jest w obu przypadkach różny. Jest to bardzo istotna